Progress Notes

Canadian Wildlife Service

S.C.F. - C.W.S.

JAN 26 1976

Progress Notes contain *interim* data and conclusions and are presented as a service to other wildlife biologists and agencies.

No. 58, January 1976

Two computer programs for an analysis of water quality data

by E.A. Driver[1] and D.C. Fowler[2]

Abstract
The rationale for using computer programs when analysing water quality data is given. The first program calculates the percent error chemical balance between anions and cations of a water sample. The second program presents a graphic display of the proportions of ions and labels this display as to date, pond number, location and concentration.

Résumé
Le rapport qui suit justifie cet emplpi-ci de programmes d'ordinateur. Le premier programme permet le calcul du pourcentage d'erreur du bilan chimique des anions et cations d'un échantillon d'eau. Le second offre une présentation visuelle des proportions d'ions comportant l'énoncé de la date du prélèvement du numéro de l'étang ainsi utilisé ainsi que de son site, et de la concentration.

Discussion
Aquatic biologists conducting water quality monitoring programs can establish the accuracy of the results of chemical analyses by determining the balance between major anions to cations as a percent error (Thomas 1953). The analytical results are first converted from milligrams per litre or parts per million to milliequavalents per litre (meq/l). The milliequivalent values of individual ions can be obtained from Thomas (1953) or Standard Methods (1965). The difference between the sum of total cation concentration, A (meq/l), and total anion concentration, B (meq/l), divided by the total ion concentration (meq/l), calculated as a percentage, is the percent error of the analysis (equation 1).

$$[1] \qquad \% \text{ error} = \frac{(A) - (B)}{(A) + (B)} \times 100$$

The precision of this calculation is set by the investigator. Many biologists consider only the following as major ions in a water sample: calcium, magnesium, sodium, potassium, chloride, bicarbonate and sulphate.

Table 1 gives an example from a water sample collected at Floral, Saskatchewan. Using equation 1 and these data, we calculate the per cent error to be 1.37.

To obtain a more complete chemical balance, soluble trace elements (copper, zinc, lead, aluminium, phosphate and ammonia) and minor constituents (iron, nitrate and fluoride) may be incorporated into the percent error calculation. Usually these trace elements are not significant enough to materially affect the overall accuracy of the test.

The percent error can be either positive or negative, the sign indicating a possible source of error. One must decide whether the analysis is acceptable or not.

The system adopted by Inland Waters Branch, Department of the Environment, for accepting or rejecting an analysis was based on criteria described by Dole and by Smith (cited in Thomas 1953). The criteria used by Smith or Dole were a range of total dissolved solids (T.D.S.). If the percent error exceeded the maximum permissible percent error for the range, the sample was rejected. We used a value for our maximum permissible error midway between the permissible errors established by Dole and by Smith as a criterion for acceptance or rejection (Table 2).

If specific conductivity had been measured instead of total dissolved solids, then it can be converted to parts per million or milligrams per litre by multiplying the conductivity by the values of the ratio of total dissolved solids to conductivity. These converted values can then be compared with values in Table 2. In Thomas (1953) the ratios vary from 0.5 to 0.7. For data collected in Saskatchewan, a ratio of 0.6 was calculated for conductivities less than 600 μmho/cm^2, ratios 0.6 to 0.75 for conductivities of 600 to 1500 μmhos/cm^2, and 0.75 to 0.83 for conductivities above 1500 μmhos/cm^2.

The computer program for the calculation of percent error was written in Fortran IV for an IBM 370 (Appendix I—Ionic Balance). Changes to the program can be made if additional ions are to be included.

An extension of this program and the data is a graphic presentation of the percent milliequivalents of anions and

Table 1
Results of analysis of water sample taken from pond G5 on 16 May 1968 at Floral, Saskatchewan.
Conductivity 1830 μmhos

	meq		mg/l		meq/l
Cations (A)					
Calcium	0.04990	X	167	=	8.3333
Magnesium	0.08224	X	50	=	4.1120
Sodium	0.04350	X	177	=	7.6995
Potassium	0.02558	X	37	=	0.9465
Total A					21.0913
Anions (B)					
Sulphate	0.02082	X	717	=	14.9279
Chloride	0.02820	X	42	=	1.1844
Bicarbonate	0.01639	X	269	=	4.4089
Total B					20.5212

[1]Prairie Migratory Bird Research Centre, CWS, Saskatoon, Sask. S7N 0X4.
[2]D.C. Fowler & Associates, 414 Fisher Cres., Saskatoon, Sask. S7L 5E3.

cations (Fig. 1). The upper panel represents cations and the lower panel anions in percent milliequivalents. The program also plots date, pond number, location and conductivity and patterns areas for calcium, magnesium, sodium, potassium, bicarbonate, sulphate and chloride.

This program is also in Fortran IV and is designed to run on the IBM 370 which utilizes Overall Plotting System designed by Stanford University with modifications by the University of Saskatchewan Computer Centre (Appendix II—Water Quality Bar Diagrams). Again, addition of other ions and data external to the plot are possible.

References

Anon. 1965. Standard methods for the examination of water and waste-water. Amer. Public Health Assoc., New York. 626 pp.

Thomas, J.F.J. 1953. Water survey report No. 1. Scope, procedure, and interpretation of survey studies. Can. Dept. Mines and Tech. Surveys. Queen's Printer. Ottawa. 69 pp.

Table 2
Criteria for rejecting or accepting analyses

Dissolved solids (ppm or mg/l) ←Range→		Maximum permissible error (%) Dole*	Smith*	CWS modification
–	50	15	22	18.5
50	100	7	10	8.5
100	200	5	7	6.0
200	500	4	6	5.0
500	1000	3	5	4.0
1000	2000	2	3	2.5
2000	greater	–	–	2.5

*Thomas 1953.

Figure 1
Print-out of water quality bar diagrams



Appendix I
Percent error determination. Ionic balance. Program listing.

```
//IONBAL    JOB (999999,'C.W.S.   ',    D.C.FOWLER
// 1,5),D.C.FOWLER,CLASS=A
//      EXEC     FORTGCLG
//FORT.SYSIN  DD   *
      REAL  POND(2),MG,NA,K
      INTEGER*2 DY,MN,YR,I
C
   1C READ(5,100,END=50) (POND(I),I=1,2),DY,MN,YR,CCND,CA,MG,NA,K,
     1                   HCO3,CL,SO4
  100 FORMAT(2A3,1X,3I2,6F8.0)
      WRITE(6,200) (POND(I),I=1,2),DY,MN,YR,CCND
      XCA=CA*0.04990
      XMG=MG*0.08224
      XNA=NA*0.04350
      XK=K*0.02558
      XHCO3=HCO3*0.01639
      XCL=CL*0.02820
      XSO4=SO4*0.02082
      SUMC=CA+MG+NA+K
      SUMA=XCA+XMG+XNA+XK
      SUMD=HCO3+CL+SO4
      SUMB=XHCO3+XCL+XSO4
      Y=SUMC-SUMD
      PERC=(SUMA-SUMB)/(SUMA+SUMB)*100.0
      WRITE(6,201) CA,XCA,MG,XMG,NA,XNA,K,XK,SUMC,SUMA
      WRITE(6,202) HCO3,XHCO3,CL,XCL,SO4,XSO4,SUMD,SUMB
      WRITE(6,203) Y
      WRITE(6,204) PERC
      GO TO 10
  200 FORMAT('1',10X,'POND: ',2A3,10X,'DATE: ',I2,'-',I2,'-',I2,10X,
     1'CONDUCTIVITY: ',F9.0/' ',16X,6('-'),16X,8('-'),24X,9('-'))
  201 FORMAT('0',38X,'PPM'/'0',12X,'CA',5X,'0.04990   X   ',F10.4,'   =   ',
     1F10.4/'0',12X,'MG',5X,'0.08224   X   ',F10.4,'   =   ',F10.4/'0',
     212X,'NA',5X,'0.04350   X   ',F10.4,'   =   ',F10.4/'0',13X,'K',5X,
     3'0.02558   X   ',F10.4,'   =   ',F10.4/' ',31X,10('-'),5X,10('-')/
     4' ',26X,'(C)',F12.4,5X,F12.4,' (A)')
  202 FORMAT('-',10X,'HCO3',5X,'0.01639   X   ',F10.4,'   =   ',F10.4/
     1'0',12X,'CL',5X,'0.02820   X   ',F10.4,'   =   ',F10.4/'0',11X,'SO4',
     25X,'0.02082   X   ',F10.4,'   =   ',F10.4/' ',31X,10('-'),5X,10('-')/
     3' ',26X,'(D)',F12.4,5X,F12.4,' (B)')
  203 FORMAT('-',10X,'Y = (C) - (D) =',F10.4,' PPM')
  204 FORMAT('-',20X,'(A) - (B)'/' ',10X,'% ERROR = --------- X 100 =',
     1 F9.4/' ',20X,'(A) + (B)')
   50 CONTINUE
      STOP
      END
/*
//GO.SYSIN  DD   *
G5     160568    1830      167       50       177      37       269        42       717
/*
```

Appendix II
Water quality bar diagrams. Program listing.

```
-----------------------------------------------------------------------------------------
//BARS      JOB (999999,'C.W.S.  ',   D.C.FOWLER
// 1,5),D.C.FOWLER,CLASS=A
//STEP2   EXEC   PLOT,DSN='A900723.WATQUAT.BAR'
//FORT.SYSIN  DD  *
C-----------------------------------------------------------------------
C
C
C                    PLOTTING WATER QUALITY BAR DIAGRAMS
C
C
C-----------------------------------------------------------------------
C
C----------DECLARE TYPES AND ARRAYS:
C
      REAL    AMODES(200),DATE(3),NUMB(2),NAME(4),MGPL(7),BOXX(7),EMP(7),
     1        BOXY(7),PERC(7),CONST(7),X1(9),X2(9),Y1(9),Y2(9),COND(2),
     2        LP,TITLE(7),BOXA(5),BOXB(5),TX1(5),TX2(5),TY1(5),TY2(5),
     3        KEY1(3),KEY2,KEY3(2),KEY4(2),KEY5(3),KEY6(4)
C
      INTEGER*2  CTR,I,K
C
C----------INITIALIZE CONSTANT DATA:
C
      DATA BOXX/2*8.0,2*28.0,2*8.0,28.0/
      DATA BOXY/10.0,2*12.0,2*8.0,2*10.0/
      DATA CONST/0.04990, 0.08224, 0.04348, 0.02558,
     1                   0.01639, 0.02082, 0.02820/
      DATA BOXA/7.0,2*29.0,2*7.0/
      DATA BOXB/2*31.0,2*29.0,31.0/
      DATA TX1/13.0,18.0,23.0,14.5,21.5/
      DATA TY1/3*12.0,2*10.0/
      DATA TX2/13.0,18.0,23.0,14.5,21.5/
      DATA TY2/3*10.0,2*8.0/
C
C----------FORMAT DICTIONARY
C
 1000 FORMAT(20A4)
 1001 FORMAT(7F10.0)
C
C***********************************************************************
C
C       INITIALIZE THE PLOTTING SYSTEM AND PRODUCE THE TITLE PAGE
C       -----------------------------------------------------
C
      CALL MODESG(AMODES,'D.FOWLER,OTHERS,999999',22)
C
C----------READ IN CONSTANT DATA USED IN PLOTTING THE
C          TITLE PAGE FROM UNIT #2
C
      READ(2,1000)  (TITLE(I),I=1,7),(KEY1(I),I=1,3),KEY2,
     1              (KEY3(I),I=1,2),(KEY4(I),I=1,2),
     2              (KEY5(I),I=1,3),(KEY6(I),I=1,4)
C
C----------SET FRAME SIZE,OBJECT SPACE,AND SUBJECT SPACE
C
      CALL SETSMG(AMODES,97,800.0)
      CALL OBJECG(AMODES,0.0,0.0,800.0,1000.0)
```

(Cont'd next page)


```
      CALL SUBJEG(AMODES,0.0,0.0,32.0,40.0)
C
C----------INCREASE CHARACTER SIZE AND PLOT THE HEADINGS
C
      CALL SETSMG(AMODES,40,12.)
      CALL SETSMG(AMODES,41,21.)
      CALL VECSG(AMODES,8.0,30.0,28,TITLE)
C
C----------REDUCE CHARACTER SIZE TO NORMAL AND ENCLOSE
C          THE HEADING IN A BOX
C
      CALL SETSMG(AMODES,40,8.)
      CALL SETSMG(AMODES,41,14.)
      CALL LINESG(AMODES,5,BOXA,BOXB)
C
C----------PLOT SAMPLE BAR DIAGRAM
C
      CALL LINESG(AMODES,7,BOXX,BOXY)
      CALL SEGMTG(AMODES,5,TX1,TY1,TX2,TY2)
C
C----------PLOT CHARACTER STRINGS EXPLAINING BAR DIAGRAM
C
      CALL VECSG(AMODES,16.0,18.0,12,KEY1)
      CALL VECSG(AMODES,8.0,14.0,4,KEY2)
      CALL VECSG(AMODES,24.0,14.0,8,KEY3)
      CALL VECSG(AMODES,8.0,6.0,8,KEY4)
      CALL VECSG(AMODES,20.0,6.0,12,KEY5)
      CALL SETSMG(AMODES,40,4.)
      CALL SETSMG(AMODES,41,7.)
      CALL VECSG(AMODES,20.0,5.0,16,KEY6)
      CALL VECTG(AMODES,23.36,5.08,12)
C
C----------PLOT CHARACTERS WITHIN THE BAR DIAGRAM
C
      CALL VECTG(AMODES,10.4,10.86,2)
      CALL VECTG(AMODES,15.4,10.86,8)
      CALL VECTG(AMODES,20.4,10.86,2)
      CALL VECTG(AMODES,11.36,8.86,52)
      CALL VECTG(AMODES,17.88,8.86,53)
      CALL SETSMG(AMODES,40,8.)
      CALL SETSMG(AMODES,41,14.)
      CALL VECTG(AMODES,10.0,11.0,4)
      CALL VECTG(AMODES,15.0,11.0,21)
      CALL VECTG(AMODES,20.0,11.0,22)
      CALL VECTG(AMODES,25.0,11.0,19)
      CALL VECTG(AMODES,10.0,9.0,9)
      CALL VECTG(AMODES,10.48,9.0,4)
      CALL VECTG(AMODES,10.96,9.0,23)
      CALL VECTG(AMODES,17.0,9.0,35)
      CALL VECTG(AMODES,17.48,9.0,23)
      CALL VECTG(AMODES,24.0,9.0,4)
      CALL VECTG(AMODES,24.48,9.0,20)
C
C***********************************************************************
C
C                    PLOTTING WATER QUALITY BAR DIAGRAMS
C                    -----------------------------------
C
C
```

(Cont'd next page)

```
C----------SET NEW FRAME SIZE
      CALL PICTRG(AMODES)
      CALL SETSMG(AMODES,97,800.0)
C
C----------READ IN DATA FOR A POND. DETERMINE WHETHER THE
C          BAR DIAGRAM SHOULD GO IN THE UPPER OR LOWER SECTION,
C          AND SET THE OBJECT SPACE AND THE SUBJECT SPACE
C               NOTE: POND DATA IS READ FROM UNIT #5
C
    4 CTR=0
    5 READ(5,1000,END=999) (DATE(I),I=1,3),(NUMB(I),I=1,2),
     1             (NAME(I),I=1,4),(COND(I),I=1,2)
      READ(5,1001) (MGPL(I),I=1,7)
      IF(CTR.EQ.1) GO TO 10
      CALL OBJECG(AMODES,0.0,500.0,800.0,1000.0)
      GO TO 11
   10 CALL OBJECG(AMODES,0.0,0.0,800.0,500.0)
   11 CTR=CTR+1
      CALL SUBJEG(AMODES,0.0,0.0,32.0,20.0)
C
C----------PLOT THE BAR STRUCTURE AND IDENTIFICATION INFORMATION
C
      CALL LINESG(AMODES,7,BOXX,BOXY)
      CALL VECSG(AMODES,8.0,14.0,12,DATE)
      CALL VECSG(AMODES,23.0,14.0,8,NUMB)
      CALL VECSG(AMODES,8.0,6.0,16,NAME)
      CALL VECSG(AMODES,23.0,6.0,8,COND)
C
C----------CALCULATE THE REQUIRED PERCENTAGES
C
      DO 12 I=1,7
      PERC(I)=0.0
   12 EMP(I)=MGPL(I)*CONST(I)
      TOPT=0.0
      DO 13 I=1,4
   13 TOPT=TOPT+EMP(I)
      BOTT=0.0
      DO 14 I=5,7
   14 BOTT=BOTT+EMP(I)
      DO 15 I=1,4
      IF(TOPT.GT.0) PERC(I)=EMP(I)/TOPT
   15 CONTINUE
      DO 16 I=5,7
      IF(BOTT.GT.0) PERC(I)=EMP(I)/BOTT
   16 CONTINUE
C
C----------SET UP ARRAYS CONTAINING THE POINTS OF THE
C          SEPARATION LINES IN THE BAR DIAGRAM AND PLOT
C          THE SEPARATING LINES
C
      X1(1)=8.0
      X1(5)=28.0
      X1(6)=8.0
      X1(9)=28.0
      TOT=0.0
      DO 40 I=1,4
      TOT=TOT+PERC(I)
   40 PERC(I)=TOT
```

(Cont'd next page)


```
      TOT=0.0
      DO 41 I=5,7
      TOT=TOT+PERC(I)
   41 PERC(I)=TOT
      DO 42 I=1,3
   42 X1(I+1)=(20.0*PERC(I))+8.0
      DO 43 I=5,6
   43 X1(I+2)=(20.0*PERC(I))+8.0
      DO 19 I=1,9
   19 X2(I)=X1(I)
      DO 20 I=1,5
      Y1(I)=12.0
   20 Y2(I)=10.0
      DO 21 I=6,9
      Y1(I)=10.0
   21 Y2(I)=8.0
      CALL SEGMTG(AMODES,9,X1,Y1,X2,Y2)
C
C----------TO SHADE SECTIONS 2,3,5,+6 OF THE BAR DIAGRAM
C          WITH HORIZONTAL LINES
C
      K=2
      ASSIGN 23 TO M
   22 A1=X1(K)
      B1=Y1(K)
      A3=X2(K)
      B3=Y2(K)
      K=K+1
      A2=X1(K)
      B2=Y1(K)
      A4=X2(K)
      B4=Y2(K)
      CALL MLTPLG(AMODES,7,A1,B1,A2,B2,A3,B3,A4,B4)
      GO TO M,(23,24,25,26)
   23 K=3
      ASSIGN 24 TO M
      GO TO 22
   24 K=6
      ASSIGN 25 TO M
      GO TO 22
   25 K=7
      ASSIGN 26 TO M
      GO TO 22
   26 CONTINUE
C
C----------TO SHADE SECTIONS 3,4,+5 OF THE BAR DIAGRAM
C          WITH VERTICAL LINES
C
      K=3
      ASSIGN 28 TO MM
   27 A1=X1(K)
      B1=Y1(K)
      A2=X2(K)
      B2=Y2(K)
      K=K+1
      A3=X1(K)
      B3=Y1(K)
      A4=X2(K)
```

(Cont'd next page)

```
      B4=Y2(K)
      NLINE=IFIX((A3-A1)/0.25+0.5)
      CALL MLTPLG(AMODES,NLINE,A1,B1,A2,B2,A3,B3,A4,B4)
      GO TO MM,(28,29,30)
   28 K=4
      ASSIGN 29 TO MM
      GO TO 27
   29 K=6
      ASSIGN 30 TO MM
      GO TO 27
   30 CONTINUE
C
C----------IF ANOTHER DIAGRAM CAN BE PLOTTED ON THIS PAGE
C          THEN DO SO, IF NOT THEN ADVANCE THE PLOTTER TO
C          A NEW PAGE
C
      IF(CTR.EQ.1) GO TO 5
      CALL PICTRG(AMODES)
      CALL SETSMG(AMODES,97,800.0)
      GO TO 4
C
C----------TERMINATE THE PLOTTER
C
  999 CALL EXITG(AMODES)
      STOP
      END
/*
//GO.FT02F001  DD  *
 WATER QUALITY BAR DIAGRAMS ** KEY **   DATE POND # LOCATIONCONDUCTIVITYMICRO MH
O @ 25 C
/*
//GO.SYSIN  DD  *
MAY 16,1968   G5         FLORAL        1,830
       167       50       177       37       269       717        42
/*
--------------------------------------------------------------------------------
```